<commit_before>dnl __BEGIN_LICENSE__
dnl Copyright (C) 2006, 2007 United States Government as represented by
dnl the Administrator of the National Aeronautics and Space Administration.
dnl All Rights Reserved.
dnl __END_LICENSE__


# Usage: AX_PKG_BOOST_CHECK_VERSION([config-prefix])
# config-prefix should be the same as was used above
AC_DEFUN([AX_PKG_BOOST_CHECK_VERSION],
[
  AC_LANG_CONFTEST(
  [AC_LANG_PROGRAM([[
#include <iostream>
#include <boost/version.hpp>
#define STR2(s) #s
#define STR(s) STR2(s)
]],[[
std::cout << STR(BOOST_VERSION);
]])])
  $CXX $OTHER_CPPFLAGS -I${PKG_BOOST_INCDIR} -o conftest conftest.$ac_ext
  BOOST_VERSION=`./conftest`
  AC_DEFINE_UNQUOTED([BOOST_VERSION],
                     [$BOOST_VERSION],
                     [The version of Boost with which the software built.])

  AH_VERBATIM([_OTHER_CHECK_BOOST_VERSION],
[// Check to make sure the user is using the same version of Boost
// headers that the software was built with.
#include <boost/version.hpp>
#if BOOST_VERSION != $1BOOST_VERSION
#error You are using a different version of Boost than you used to build the software!
#endif
])
])<commit_msg>new version of boost version macro<commit_after>dnl __BEGIN_LICENSE__
dnl Copyright (C) 2006, 2007 United States Government as represented by
dnl the Administrator of the National Aeronautics and Space Administration.
dnl All Rights Reserved.
dnl __END_LICENSE__


# Usage: AX_PKG_BOOST_CHECK_VERSION([config-prefix])
# config-prefix should be the same as was used above
AC_DEFUN([AX_PKG_BOOST_CHECK_VERSION],
[
  boost_check_save_CPPFLAGS="$CPPFLAGS"
  boost_check_save_LDFLAGS="$LDFLAGS"

  CPPFLAGS="$PKG_BOOST_CPPFLAGS"
  LDFLAGS="$PKG_BOOST_LDFLAGS"

  AC_RUN_IFELSE(
    [AC_LANG_PROGRAM([[
      #include <fstream>
      #include <boost/version.hpp>
    ]], [[
      std::ofstream out("conftest.out");
      if (!out.is_open())
          return 1;
      out << BOOST_VERSION;
    ]])],
    [BOOST_VERSION=`cat conftest.out`],
    [AC_MSG_ERROR([Failed to run Boost version check program])])

  if test -z "$BOOST_VERSION"; then
    AC_MSG_ERROR([Boost version check program did not define BOOST_VERSION])
  fi

  CPPFLAGS="$boost_check_save_CPPFLAGS"
  LDFLAGS="$boost_check_save_LDFLAGS"

  AC_DEFINE_UNQUOTED([BOOST_VERSION],
                     [$BOOST_VERSION],
                     [The version of Boost with which the software built.])

  AH_VERBATIM([_OTHER_CHECK_BOOST_VERSION],
[// Check to make sure the user is using the same version of Boost
// headers that the software was built with.
#include <boost/version.hpp>
#if BOOST_VERSION != $1BOOST_VERSION
#error You are using a different version of Boost than you used to build $1!
#endif
])
])
{'fraction_non_alphanumeric': 0.09671465485418974, 'fraction_numerical': 0.008121077888519748, 'mean_word_length': 4.313725490196078, 'pattern_counts': {'":': 0, '<': 13, '<?xml version=': 0, '>': 9, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '5979758', 'n_tokens_mistral': 933, 'n_tokens_neox': 846, 'n_words': 258}